
NAG Library Routine Document

F01KCF

Note: before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

F01KCF computes an estimate of the absolute condition number of a matrix function f of a complex n
by n matrix A in the 1-norm, using analytical derivatives of f you have supplied.

2 Specification

SUBROUTINE F01KCF (N, A, LDA, F, IUSER, RUSER, IFLAG, CONDA, NORMA,
NORMFA, IFAIL)

&

INTEGER N, LDA, IUSER(*), IFLAG, IFAIL
REAL (KIND=nag_wp) RUSER(*), CONDA, NORMA, NORMFA
COMPLEX (KIND=nag_wp) A(LDA,*)
EXTERNAL F

3 Description

The absolute condition number of f at A, condabs f; Að Þ is given by the norm of the Fréchet derivative of
f , L Að Þ, which is defined by

L Xð Þk k :¼ max E 6¼0
L X;Eð Þk k

Ek k ;

where L X;Eð Þ is the Fréchet derivative in the direction E. L X;Eð Þ is linear in E and can therefore be
written as

vec L X;Eð Þð Þ ¼ K Xð Þvec Eð Þ;

where the vec operator stacks the columns of a matrix into one vector, so that K Xð Þ is n2 � n2. F01KCF
computes an estimate � such that � � K Xð Þk k1, where K Xð Þk k1 2 n�1 L Xð Þk k1; n L Xð Þk k1

� �
. The

relative condition number can then be computed via

condrel f; Að Þ ¼ condabs f; Að Þ Ak k1

f Að Þk k1

:

The algorithm used to find � is detailed in Section 3.4 of Higham (2008).

4 References

Higham N J (2008) Functions of Matrices: Theory and Computation SIAM, Philadelphia, PA, USA

5 Parameters

1: N – INTEGER Input

On entry: n, the order of the matrix A.

Constraint: N � 0.

2: AðLDA; �Þ – COMPLEX (KIND=nag_wp) array Input/Output

Note: the second dimension of the array A must be at least N.

On entry: the n by n matrix A.

On exit: the n by n matrix, f Að Þ.

3: LDA – INTEGER Input

On entry: the first dimension of the array A as declared in the (sub)program from which F01KCF
is called.

Constraint: LDA � N.

4: F – SUBROUTINE, supplied by the user. External Procedure

Given an integer m, the subroutine F evaluates f mð Þ zið Þ at a number of points zi.

The specification of F is:

SUBROUTINE F (M, IFLAG, NZ, Z, FZ, IUSER, RUSER)

INTEGER M, IFLAG, NZ, IUSER(*)
REAL (KIND=nag_wp) RUSER(*)
COMPLEX (KIND=nag_wp) Z(NZ), FZ(NZ)

1: M – INTEGER Input

On entry: the order, m, of the derivative required.

If M ¼ 0, f zið Þ should be returned. For M > 0, f mð Þ zið Þ should be returned.

2: IFLAG – INTEGER Input/Output

On entry: IFLAG will be zero.

On exit: IFLAG should either be unchanged from its entry value of zero, or may be set
nonzero to indicate that there is a problem in evaluating the function f zð Þ; for instance
f zð Þ may not be defined. If IFLAG is returned as nonzero then F01KCF will terminate
the computation, with IFAIL ¼ 3.

3: NZ – INTEGER Input

On entry: nz, the number of function or derivative values required.

4: ZðNZÞ – COMPLEX (KIND=nag_wp) array Input

On entry: the nz points z1; z2; . . . ; znz at which the function f is to be evaluated.

5: FZðNZÞ – COMPLEX (KIND=nag_wp) array Output

On exit: the nz function or derivative values. FZðiÞ should return the value f mð Þ zið Þ, for
i ¼ 1; 2; . . . ; nz.

6: IUSERð�Þ – INTEGER array User Workspace
7: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

F is called with the parameters IUSER and RUSER as supplied to F01KCF. You are free
to use the arrays IUSER and RUSER to supply information to F as an alternative to
using COMMON global variables.

F must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)program
from which F01KCF is called. Parameters denoted as Input must not be changed by this
procedure.

5: IUSERð�Þ – INTEGER array User Workspace
6: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

IUSER and RUSER are not used by F01KCF, but are passed directly to F and may be used to
pass information to this routine as an alternative to using COMMON global variables.

7: IFLAG – INTEGER Output

On exit: IFLAG ¼ 0, unless IFLAG has been set nonzero inside F, in which case IFLAG will be
the value set and IFAIL will be set to IFAIL ¼ 3.

8: CONDA – REAL (KIND=nag_wp) Output

On exit: an estimate of the absolute condition number of f at A.

9: NORMA – REAL (KIND=nag_wp) Output

On exit: the 1-norm of A.

10: NORMFA – REAL (KIND=nag_wp) Output

On exit: the 1-norm of f Að Þ.

11: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this parameter you should
refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit (as
defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

An internal error occurred when estimating the norm of the Fréchet derivative of f at A. Please
contact NAG.

IFAIL ¼ 2

An internal error occurred when evaluating the matrix function f Að Þ. You can investigate further
by calling F01FMF with the matrix A and the function f .

IFAIL ¼ 3

IFLAG has been set nonzero by the user-supplied subroutine.

IFAIL ¼ �1

On entry, N < 0.

IFAIL ¼ �3

On entry, parameter LDA is invalid.
Constraint: LDA � N.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.8 in the Essential Introduction for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.7 in the Essential Introduction for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.6 in the Essential Introduction for further information.

7 Accuracy

F01KCF uses the norm estimation routine F04ZDF to estimate a quantity �, where � � K Xð Þk k1 and
K Xð Þk k1 2 n�1 L Xð Þk k1; n L Xð Þk k1

� �
. For further details on the accuracy of norm estimation, see the

documentation for F04ZDF.

8 Parallelism and Performance

F01KCF is threaded by NAG for parallel execution in multithreaded implementations of the NAG
Library. In these implementations, this routine may make calls to the user-supplied functions from within
an OpenMP parallel region. Thus OpenMP directives within the user functions can only be used if you
are compiling the user-supplied function and linking the executable in accordance with the instructions
in the Users’ Note for your implementation. The user workspace arrays IUSER and RUSER are
classified as OpenMP shared memory and use of IUSER and RUSER has to take account of this in order
to preserve thread safety whenever information is written back to either of these arrays. If at all possible,
it is recommended that these arrays are only used to supply read-only data to the user functions when a
multithreaded implementation is being used.

F01KCF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor
library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users’ Note for your
implementation for any additional implementation-specific information.

9 Further Comments

Approximately 6n2 of complex allocatable memory is required by the routine, in addition to the memory
used by the underlying matrix function routine F01FMF.

F01KCF returns the matrix function f Að Þ. This is computed using F01FMF. If only f Að Þ is required,
without an estimate of the condition number, then it is far more efficient to use F01FMF directly.

The real analogue of this routine is F01JCF.

10 Example

This example estimates the absolute and relative condition numbers of the matrix function e3A where

A ¼
1:0þ 1:0i 0:0þ 1:0i 1:0þ 0:0i 2:0þ 0:0i
0:0þ 0:0i 2:0þ 0:0i 0:0þ 2:0i 1:0þ 0:0i
0:0þ 1:0i 0:0þ 1:0i 0:0þ 0:0i 2:0þ 0:0i
1:0þ 0:0i 0:0þ 1:0i 1:0þ 0:0i 0:0þ 1:0i

0
B@

1
CA:

10.1 Program Text

! F01KCF Example Program Text
! Mark 25 Release. NAG Copyright 2014.

Module f01kcfe_mod

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Accessibility Statements ..
Private
Public :: fexp3

Contains
Subroutine fexp3(m,iflag,nz,z,fz,iuser,ruser)

! .. Scalar Arguments ..
Integer, Intent (Inout) :: iflag
Integer, Intent (In) :: m, nz

! .. Array Arguments ..
Complex (Kind=nag_wp), Intent (Out) :: fz(nz)
Complex (Kind=nag_wp), Intent (In) :: z(nz)
Real (Kind=nag_wp), Intent (Inout) :: ruser(*)
Integer, Intent (Inout) :: iuser(*)

! .. Intrinsic Procedures ..
Intrinsic :: cmplx, exp

! .. Executable Statements ..
Continue
fz(1:nz) = (cmplx(3.0E0_nag_wp,0.0_nag_wp,kind=nag_wp)**m)* &

exp((3.0E0_nag_wp,0.0E0_nag_wp)*z(1:nz))
! Set iflag nonzero to terminate execution for any reason.

iflag = 0
Return

End Subroutine fexp3
End Module f01kcfe_mod
Program f01kcfe

! F01KCF Example Main Program

! .. Use Statements ..
Use nag_library, Only: f01kcf, nag_wp, x02ajf, x04daf
Use f01kcfe_mod, Only: fexp3

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6

! .. Local Scalars ..
Real (Kind=nag_wp) :: conda, cond_rel, eps, norma, &

normfa
Integer :: i, ifail, iflag, lda, n

! .. Local Arrays ..
Complex (Kind=nag_wp), Allocatable :: a(:,:)
Real (Kind=nag_wp) :: ruser(1)
Integer :: iuser(1)

! .. Executable Statements ..
Write (nout,*) ’F01KCF Example Program Results’
Write (nout,*)
Flush (nout)

! Skip heading in data file
Read (nin,*)
Read (nin,*) n

lda = n
Allocate (a(lda,n))

! Read A from data file
Read (nin,*)(a(i,1:n),i=1,n)

! Display A

ifail = 0
Call x04daf(’G’,’N’,n,n,a,lda,’A’,ifail)

! Find absolute condition number estimate
ifail = 0
Call f01kcf(n,a,lda,fexp3,iuser,ruser,iflag,conda,norma,normfa,ifail)

If (ifail==0) Then
! Print solution

Write (nout,*)
Write (nout,*) ’F(A) = exp(3A)’
Write (nout,99999) ’Estimated absolute condition number is: ’, conda

! Find relative condition number estimate
eps = x02ajf()
If (normfa>eps) Then

cond_rel = conda*norma/normfa
Write (nout,99999) ’Estimated relative condition number is: ’, &

cond_rel
Else

Write (nout,99998) ’The estimated norm of f(A) is effectively zero’, &
’and so the relative condition number is undefined.’

End If

End If

99999 Format (1X,A,F7.2)
99998 Format (/1X,A/1X,A)

End Program f01kcfe

10.2 Program Data

F01KCF Example Program Data

4 :Value of N

(1.0, 1.0) (0.0, 1.0) (1.0, 0.0) (2.0, 0.0)
(0.0, 0.0) (2.0, 0.0) (0.0, 2.0) (1.0, 0.0)
(0.0, 1.0) (0.0, 1.0) (0.0, 0.0) (2.0, 0.0)
(1.0, 0.0) (0.0, 1.0) (1.0, 0.0) (0.0, 1.0) :End of matrix A

10.3 Program Results

F01KCF Example Program Results

A
1 2 3 4

1 1.0000 0.0000 1.0000 2.0000
1.0000 1.0000 0.0000 0.0000

2 0.0000 2.0000 0.0000 1.0000
0.0000 0.0000 2.0000 0.0000

3 0.0000 0.0000 0.0000 2.0000
1.0000 1.0000 0.0000 0.0000

4 1.0000 0.0000 1.0000 0.0000
0.0000 1.0000 0.0000 1.0000

F(A) = exp(3A)
Estimated absolute condition number is: 9474.43
Estimated relative condition number is: 13.74

